
超幾何関数とJのコード解析およびその利用

西川　利男

１．ガウスの超幾何関数とは[1]

　数学の王者、フリードリッヒ・ガウスは1812年に次の超幾何微分方程式

0))1(()1( =+′−+++′′− α β γγβα yxyxx

の解として得られる、次のようなべき級数で表される超幾何(Hypergeometric)関数を

発表した。

  これは、解析的操作に熟練するため、取り上げられ、やがて忘れられてしまうよう

な、単なる教科書演習問題ではなかった。パラメータα、β、γにある特定な値を与

えると、例えば、対数、三角関数などから始まり、数理物理学に現れるさまざまな関

数となるのである。

　そのいくつかを挙げてみる。

２．Jの超幾何関数(H.)の定義コードの解析

２．１　接続詞(!.)と拡張された階乗

　はじめに、超幾何関数の定義で重要な役割を持つ接続詞(!.)についてすこし述べる。

接続詞(!.)(Customize)は 2つの特定の引数をとり、通常にはない新しい機能を持った

プリミティブを作成する(Customize)。これについては以前に報告したことがある[2]

が、今回は別の機能である。

-----------------------------------------------------------------------------

[1] ベル、田中、銀林訳「数学をつくった人びと」上、p.265　東京図書(1985).

[2] 西川利男「Jの接続詞 !. Fit(Customize)とは」JAPLA研究会資料 2009/4/25

　ここでは、次のように定義されている。

　x ^ !. r y ←→ */ x + r * i. y

つまり、接続詞(!.)の左引数として動詞 ^ 、右引数として名詞の値 rをとり、新し

く２項動詞を作る。そしてその動詞の左引数としてx、右引数としてyをとる。

　この使い方を見るために、例えば次のような簡単な実験をしてみる。

　5 (^ !. _1) 3  → */5 + _1 * i.3 →  */5 + _1 * 0 1 2 → */5 + 0 _1 _2

                 → */5 4 3  → 60

つまり、 4 から3だけ下降する拡張された階乗(falling factorial)が得られる。

  あるいは、次のようになる。　　

　5 (^ !. 1) 3  → */5 + 1 * i.3 →  */5 + 1 * 0 1 2 → */5 + 0 1 2

                → */5 6 7  → 210

この場合は4から3だけ上昇する拡張された階乗(Rising factorial)が得られる。

　これに名前をつけて定義する。

　g =: ^ !. 1

  4!:0 <’g’ → 3(動詞)

　5 g i.4 → 1 5 30 210

２．２　超幾何関数(H)の定義とその解析

　J Vocabulary の Hypergeometricには次の定義がある。

----------------------------------------------------------

rf=: 1 : '(,x.)"_ ^!.1/ i.@[' NB. Rising factorial

L1=: 2 : 'x.rf %&(*/) y.rf'

L2=: (i.@[ ^~ ]) % (!@i.@[)

Hyp=: L1 (+/ . *) L2

H =: 2 : '(m. Hyp n. ) "0'

-----------------------------------------------------------

　超幾何関数の計算を大きく見てみると、級数の各項は係数部分(L1)と xのべき乗

(L2)部分から成っていることがわかる。

　H = ∑ L1 * L2

　Vocabulary の解説には次の例が挙げられている。この例を使ってコーディングをト

レースしてみよう。

------------------------------------------------------------

  'a b'=: 2 3 5; 6 5 

  a L1 b

(2 3 5"_ ^!.1/ i.@[) %&(*/) 6 5"_ ^!.1/ i.@[

  t=: 4 [ z=: 7

  t a L1 b z

1 1 1.71429 4.28571

  t (a H b , a H. b) z

295 295

------------------------------------------------------------

　まず、係数部分L1について検討する。これは拡張された階乗の比から成っている。

　2 3 5 L1 6 5

　先に定義したgはVocabularyの定義rfと同じである。一つずつやってみる。

   2 g i.4

1 2 6 24

   3 g i.4

1 3 12 60

   5 g i.4

1 5 30 210

   6 g i.4

1 6 42 336

   

   2 3 5 g / i.4

1 2  6  24

1 3 12  60

1 5 30 210

   */ 2 3 5 g / i.4

1 30 2160 302400

   6 5 g / i.4

1 6 42 336

1 5 30 210

   */ 6 5 g / i.4

1 30 1260 70560

   (*/ 2 3 5 g / i.4) % (*/ 6 5 g / i.4)

1 1 1.71429 4.28571

　次にべき乗部分L2を見てみる。

   L2

+------------------+-+--------------+

|+--------+-----+-+|%|+--------+-+-+|

||+--+-+-+|+-+-+|]|| ||+-+-+--+|@|[||

|||i.|@|[|||^|~|| || |||!|@|i.|| | ||

||+--+-+-+|+-+-+| || ||+-+-+--+| | ||

|+--------+-----+-+| |+--------+-+-+|

+------------------+-+--------------+

Vocabulary の例のように

　4 L2 z

とすると、これは

  (i.4)^~z) % (! i.4)

となるので

　((z^0) (z^1) (z^2) (z^3)) % ((!0) (!1) (!2) (!3))

となる。

　つまり通常の階乗の逆数を係数とするべき乗である。

　最終の結果は

　
2110

!3
28571.4

!2
71429.1

!1
1

!0
1 zzzz ∗+∗+∗+∗  　

において、z = 7とした値が得られる。

　 4 (2 3 5 H 6 5) 7

295

   この機能を持つものが、超幾何関数の接続詞(H.)であり、次のように同じ結果が得

られる。

   4 (2 3 5 H. 6 5) 7

295

   　   

３．Jの超幾何関数(H.)の利用　   

　パラメータを次のようにとることにより超幾何関数は指数関数の近似になる。最初

の8項をとった場合は次のようになる。

   8 (1 H. 1) i. 6

1 2.71825 7.38095 19.8464 51.8063 128.619

　左引数をとらずに１項関数として使用したときは、級数和を収束するまで行った値

を与える。

   (1 H. 1) i. 6

1 2.71828 7.38906 20.0855 54.5982 148.413

つまり、普通の指数関数と同じ値となる。

   ^ i. 6

1 2.71828 7.38906 20.0855 54.5982 148.413

　誤差関数、正規分布関数への利用はつぎのようになる。

   erf   =: (1 H. 1.5)@*: * 2p_0.5&* % ^@:*:    NB. error function
   n01cdf=: -: @: >: @: erf @: ((%:0.5)&*)      NB. CDF of normal 0,1

   erf 0.5 1 1.5
0.5205 0.842701 0.966105
   n01cdf _2 _1.5 _1 _0.5 0 0.5 1 1.5 2
0.0227501 0.0668072 0.158655 0.308538 0.5 0.691462 0.841345 0.933193 
0.97725

- 4 -



NB. HyperGeometric Function by Gauss

rf=: 1 : '(,x.)"_ ^!.1/ i.@[' NB. Rising factorial

L1=: 2 : 'x.rf %&(*/) y.rf'

L2=: (i.@[ ^~ ]) % (!@i.@[)

Hyp=: L1 (+/ . *) L2

H =: 2 : '(m. Hyp n. ) "0'

NB.   4 (2 3 5 H 6 5) 7

NB. 295

NB.   8 (1 H 1) i.6

NB. 1 2.71825 7.38095 19.8464 51.8063 128.619

   

NB. 'a b'=: 2 3 5; 6 5 

NB. a L1 b

NB. (2 3 5"_ ^!.1/ i.@[) %&(*/) 6 5"_ ^!.1/ i.@[

NB. t=: 4 [ z=: 7

NB. t a L1 b z

NB. 1 1 1.71429 4.28571

NB. t (a H b , a H. b) z

NB. 295 295

NB. f=: 1 H. 1

NB. 8 f i. 6

NB. 1 2.71825 7.38095 19.8464 51.8063 128.619

NB. f i. 6

NB. 1 2.71828 7.38906 20.0855 54.5982 148.413

NB. ^ i. 6

NB. 1 2.71828 7.38906 20.0855 54.5982 148.413

NB. from J Vocaburary ==========================

NB. Extended Factorial / Rising

g =: ^ !. 1

g2 =: 2 g i.4

g3 =: 3 g i.4

g5 =: 5 g i.4

g6 =: 6 g i.4

h1 =: i.@[ ^ ~ ]

